
Introduction to Maple

A.1 Getting Started
Maple is a powerful mathematical calculator, often called a computer algebra system

(CAS) or symbolic mathematics program, because unlike most calculators it can do alge-

bra and calculus symbolically. For example, it can expand the product (x + 1)30 into the

normal representation of this polynomial, and it can factor the result back into the original

form. It can compute the derivative or antiderivative of 3x/
√

x2 + 2x + 47. It can also

serve as a numerical calculator, but again unlike most calculators it can do integer and

rational number arithmetic exactly, no matter how many digits are involved. If it is unde-

sirable or impossible to express a result exactly, Maple can do numerical approximation.

Finally, Maple can do some quite sophisticated graphing in both 2 and 3 dimensions.

To learn how to use some of Maple’s features, you should work through this tutorial

step by step. What you should type will be shown in the left hand column in typewriter

style; comments and explanations will appear at the right. You should do everything

listed here, but you should feel free to experiment too. If you leave something out or do

something extra it may change the results of some later computations. (If you want to

experiment along the way, try using different names for functions and variables than the

ones in the examples.)

To start Maple, log in on one of the workstations in the math lab. Click on the Maple

button on the bar at the bottom of the screen. When the Maple window comes up, you

are ready to give commands to Maple.

For the most part, the commands you give will be typed in the main Maple window.

Some commands can be given by making selections from menus, or by clicking on buttons

on the top bar. For example, to leave Maple, you may move the mouse pointer to the

button titled ‘File’ and click on the left mouse button. When you do, a list of further

commands appears. If you click on the last command, ‘Exit’, Maple will quit and the

window will disappear.

The instructions given here are for Maple version 9. To make Maple 10 act like Maple

9, choose Options from the Tools menu; choose the Display tab; change Input display

to Maple Notation; click on Apply Globally. You should only have to do this the first

time you use Maple.A.2 Algebra
A.2.1 Numbers

2+3;

123456789* 987654321;

You must type the semicolon at the end of

the line to signal the end of your expression.

2+3

34*98

-45;

Type this as is, without the semicolon on

the first line. Maple will warn you about

the missing semicolon, but will usually do

the calculation anyway. Sometimes you may

want to start a new line as part of the same

expression; you can do this by holding the

shift key while you press return. Do this

after the “98”.

2/5 + 1/3;

23/45-167/345*2/3;

All rational arithmetic is done exactly.

2^5;

2**5;

3^27;

ifactor(%);

Maple knows standard mathematical no-

tation, including these two symbols for ex-

ponentiation. The percent sign (%) denotes

the previous result (it’s called the ‘ditto’ op-

erator). The command ifactor does inte-

ger factoring; the command factor factors

polynomials.

12345/98765;

evalf(%);

You can force Maple to approximate values

as ‘floating point numbers’ (i.e., using deci-

mal notation). Some of Maple’s commands

do this automatically, as we will see later.

I^2;

evalf(Pi);

evalf(exp(1));

The number i =
√
−1 is denoted by I; the

case is significant—a lower case ‘i’ is differ-

ent. Pi denotes π; again case is significant.

It is very easy to mistakenly type ‘pi’ in-

stead of ‘Pi’. When you do, things will al-

most certainly not work properly, but you

will probably get no error messages—you

just have to watch out for this. There is no

built-in symbol for e ≈ 2.71828; use the

function exp(x) whenever you need ex.

A.2.2 Variables and Expressions

Maple can manipulate expressions involving variables as well as numbers. Variables can

be assigned values, either actual numeric values or entire expressions.

(x+5)^2;

f:=%;

x:=5;

f;

x^2;

Maple simply repeats the first line since x

has no value. The second line saves the ex-

pression (x + 5)2 with name f ; the “:=”

symbol acts like an equal sign in normal

mathematics or in the programming lan-

guage C. The third line assigns x the value

5.

x:=’x’;

x;

f;

This clears the value of x—this is not the

same as setting it to zero! Note that Maple

remembers the ‘real’ definition of f—it is

still an expression involving x, not the fixed

value 100.

2x;

2*x;

You must always type the multiplication

symbol—Maple will not accept the first line

even though it is standard notation. If you

make a simple typing mistake like this, you

can fix it without retyping the command.

In this case, Maple should correctly guess

where the problem is and place the cursor

between the 2 and the x; simply type the

* and press return. In any case, you can

always move the mouse pointer to the po-

sition of the mistake and click the left but-

ton, then make your corrections. You can

adjust your position in the line using the ar-

row keys to the right of the main keyboard.

When you press the return key, the com-

mand will be re-evaluated.

expand(f);

factor(%);

factor(x^2+4*x+5);

If a polynomial doesn’t factor, Maple leaves

it alone. (As a general rule, if Maple cannot

perform a calculation it will simply restate

the command.)

WARNING: You can go back anywhere in your Maple session and change a command.

When you do this you change only the result of that command—you do not change any

subsequent commands that depend on the result. If you want to re-execute subsequent

commands, you must move the cursor to each command in turn and press the return key.

Do not overuse this editing ability—it can make the chain of events very difficult to follow.

It often is better to copy the command, paste it in at the end of the session, modify it

there, and execute it as a new command. To copy and paste a command, do this: Position

the mouse pointer at the beginning of the command you want to copy, press the left mouse

button and hold it down, move the mouse cursor to the end of the command, and release

the mouse button. (If the command is on a single line, you may also select the whole

command by positioning the cursor in the line, and clicking the left mouse button three

times quickly—this is a “triple click”.) The command should now be highlighted. Position

the mouse pointer where you want to paste and click the middle button.

A.2.3 Evaluation and Substitution

It usually is not a good idea to assign values to variables—they are more useful as variables.

If you want Maple to evaluate an expression you can give the variables temporary values.

f:=A*(x+3)^3;

subs(A=3,f);

subs({x=5,A=2},f);

x; A; f;

Note here that you use a plain ‘=’ inside

the substitution command. Two or more

substitutions are done by enclosing them in

curly braces. The values of f , A and x don’t

change.

subs(x=z^2,f);

f:=subs(x=z^3,f);

f;

You can substitute whole expressions for

variables, not just values. To force the value

of f to change, just reassign it—but be

careful. It usually is better to give the ex-

pression a new name in case you need the

original definition of f later.

f:=x^2+3*x+4;

f(1);

g:=x -> x^2+3*x+4;

g(1);

h:=unapply(f,x);

h(1);

h(x);

h(x^3);

Maple makes a crucial distinction between

an expression like x2 + 3x + 4 and the

function which assigns to each value of x

the value x2 + 3x + 4. In this example,

f is an expression, g and h are functions.

The same distinction exists in mathemat-

ics, but usually it is possible to ignore it,

and indeed in many instances Maple will

accept either a function or an expression.

To employ the normal mathematical nota-

tion f(1), however, f must be a function.

This example shows two ways to define a

function—directly, or using an existing ex-

pression. The last two examples show that

you can evaluate a function on expressions,

not just numbers.

map(g,[-3,-2,-1,0,1,2,3]);

map(g,{-3,-2,-1,0,1,2,3});

Using the ‘map’ operation and a function

(not an expression) you can evaluate a func-

tion on a list of values. The list can be en-

closed in square brackets or in set braces—

but notice the difference! The set braces

produce a set as a result, meaning that re-

peated values are shown only once, and the

order does not necessarily correspond to the

order of the input values. In general, the

square bracket form is more useful.

A.2.4 Solving Equations

Maple will solve a limited class of equations exactly and a broader class approximately.

solve(x^2-x-10=10,x);

solve(x^2-x-20,x);

solutions:=solve(g(x),x);

solve(x^5+4*x^3-3*x^2+10,x);

solve(tan(x)=x+1,x);

allvalues(%);

solve(tan(x)=x,x);

allvalues(%);

solve(sin(x)=cos(x)-1,x);

The solve command will give exact answers

if possible, including complex answers. If

Maple can’t solve an equation it may print

nothing, or it may display the equation in a

standard form, preceded by ‘RootOf’. You

can ask Maple to attempt to evaluate roots

displayed as a RootOf using allvalues.

Maple will attempt to find all solutions for

an equation or a RootOf, but often will not

succeed.

Do not rely on the solve command alone

when looking for solutions to equations—at

the very least, you should graph your func-

tions so that you know where roots appear

to be. Note that if you give an expression

instead of an equation Maple assumes that

you mean ‘= 0’.

map(g,[solutions]);

simplify(%);

In some cases, Maple will not automatically

simplify as much as it can, so you need to

nudge it a bit. In this case, of course, we

knew that the values should be zero. If the

result had not simplified to zero, careful

scrutiny of the problem would be called for.

f:=x^3+3*x^2+x+5:

solve(f,x);

fsolve(f,x);

fsolve(f,x,complex);

You can end a line with a colon instead of

a semicolon—it tells Maple not to display

the result. This can be particularly handy

when the result is very long, or simply when

you either know what it is or do not need

to know (as when you will be using the re-

sult to compute a further result). The fsolve

operation tells Maple to approximate the

roots. In this example, Maple can compute

the roots exactly, but it may be more in-

formative to see approximations. Note that

Maple only computes real roots unless the

‘complex’ option is present. If the function

is not a polynomial, fsolve will find at most

one solution, but for polynomials it will find

all solutions.

g:=x -> exp(x)+x^2-10*x;

fsolve(g(x),x);

fsolve(g(x),x,-2..2);

g(%);

fsolve(g(x),x,2..4);

g(%);

You can specify a range in fsolve—you

would normally do this when you know

there are multiple roots and you want to

approximate a particular one. In this case,

a graph of the function clearly indicates a

root near 0 and a root near 3. Because g

is a function, and fsolve expects an expres-

sion, it is necessary to use g(x) in fsolve. (I

made g a function so that it would be easy

to check the answers.)

solve({3*x+5*y=6,4*x-6*y=-8},

{x,y});

h1:=3*x-6*y-5;

h2:=-4*x*y+7;

solve({h1,h2},{x,y});

evalf(%);

fsolve({h1,h2},{x,y});

fsolve({h1,h2},{x,y},{x=-2..0,y=-2..0});

Maple can solve equations simultaneously,

with either solve or fsolve. As with single

equations, fsolve may not find all solutions,

but you can specify ranges for x and y to

force fsolve to find particular solutions.

A.3 Plotting
Maple can do two and three dimensional graphing. There are many ways to adjust the

plots that Maple produces; we’ll cover just the basics here.

plot(f,x=-10..10);

plot(g(x),x=-5..5);

plot(g,-10..10);

plot(g,-10..10,y=-10..10);

Maple will accept either an expression or a

function to plot, but notice that the form of

the range is a bit different in the two cases.

For an expression, use ‘x=’ before the range,

but for a function leave it out. You can also

specify the range for y explicitly. This is

often useful if the function gets very large

on the interval.

You can change the appearance of the plot

in a variety of ways. Click in the plot and

you should see a border appear around it;

then you can explore the items on the plot

menu. If you prefer, you can have plots

come up in separate windows; use the Op-

tions item on the Tools menu in the main

Maple window, then open the Display tab

and change the plot display.

plot([1+2*sin(t),t,t=-Pi..Pi],

coords=polar);

You can do plots in polar coordinates. In-

side the square brackets, the first expression

is the radius, the second is the angle, and

the third is the range of the variable. In

this case, the angle and the variable are the

same, but in general the angle could be a

more complicated function of the variable.

plot({x^2,x^3},x=-5..5);

plot({

[1+2*sin(t),t,t=-Pi..Pi],

[cos(4*t),t,t=-Pi..Pi]},

coords=polar);

You can put more than one curve in the

same plot by enclosing a list of expressions

in curly braces.

r:=sqrt(x^2+y^2);

f:=sin(r)/r;

plot3d(f,x=-10..10,

y=-10..10);

Try changing the appearance of a 3D plot

by using the Style, Color, Axes and Pro-

jection menus. If you position the cursor in

the drawing area and press and hold the left

mouse button, you can move the mouse to

change the orientation of the plot—try it.

with(plots):

polarplot(1+sin(t));

polarplot(sin(t/3));

polarplot(sin(t/3),

t=0..3*Pi);

When you start Maple, only the most ba-

sic plotting commands are available. By

typing “with(plots)” you tell Maple to

load a number of special purpose plot-

ting functions. One of the most useful is

the polarplot routine shown here. By de-

fault, Maple uses the range [−π, π]; the last

command shows how to specify a different

range.

ulus
A.4.1 Limits

f:=x^2+4*x+30:

limit(f,x=3);

subs(x=3,f);

g:=sin(x)/x;

limit(g,x=0);

limit(g,x=infinity);

g:=x/abs(x);

limit(g,x=0);

limit(g,x=0,left);

limit(g,x=0,right);

The limit command limit(f,x=a) com-

putes the limit limx→a f(x). Maple can

handle many non-trivial limit problems and

limits at infinity.

A.4.2 Differentiation

f:=x->x^3+4*x^2+3*x+4:

fp:=diff(f(x),x);

critpts:=solve(fp,x);

map(f,[critpts]);

evalf(%);

evalf(critpts);

plot({f(x),fp},x=-5..5,

y=-5..10);

The diff command does differentiation.

Here I have found the derivative with re-

spect to x, then the x coordinates of the

critical points of f , then the value of f at

these points, then floating point represen-

tations for the x and y coordinates. Be-

cause Maple can do calculations exactly,

you should do approximations at the end;

it would not be a good idea to approximate

critpts before substitution into f (though in

this simple case the results are almost iden-

tical). Note that if you click in the plot, the

coordinates of the point under the cursor

appear at the upper left of the Maple win-

dow. Youe can use this to approximate crit-

ical points and double check the computed

values.

A.4.3 Implicit Differentiation

eqn:=x^2+y^2=1;

xp:=implicitdiff(eqn,x,y);

yp:=implicitdiff(eqn,y,x);

subs({x=1/2,y=sqrt(3/4)},yp);

Notice that the order of the second two ar-

guments to the implicitdiff function is

important—xp is dx/dy and yp is dy/dx.

The last line computes y′ at a specific point,

(1/2,
√

3/2).

A.4.4 Integration

f:=(2*z-3)*sqrt(z-3);

g:=int(f,z);

diff(g,z);

factor(%);

The int command shown here will compute

an antiderivative. The computer is not in-

fallible, and you may not notice a typing

mistake, so it’s not a bad idea to use Maple

to check your work. Here, I’ve taken the

derivative to check the antiderivative.

int(f,z=5..10);

simplify(%);

evalf(%);

subs(z=10,g)-subs(z=5,g);

To do definite integration give the limits af-

ter the variable. Of course, you can also do

it yourself using the antiderivative. Notice

that the answer can look quite different be-

fore simplification.

f:=1/sqrt(x):

int(f,x=0..1);

f:=1/x^2:

int(f,x=0..1);

int(f,x=1..infinity);

f:=exp(-x^2):

int(f,x=-infinity..infinity);

int(sin(x),x=0..infinity);

Maple can do many improper integrals ex-

actly. If the integral does not converge,

Maple may answer with ‘∞’ if appropriate,

or it may give an error message.

f:=tan(x^2):

int(f,x);

int(f,x=0..1);

evalf(%);

Int(f,x=0..1);

evalf(%);

Maple can’t find antiderivatives of some

functions. In such cases you can use evalf

to approximate the value of a definite inte-

gral. If you don’t want Maple to even try to

find an antiderivative, use the capital I form

of the integration operator.

You can add plain text to a Maple session. This is especially useful if you are going to give

a printed or electronic version of a session to someone else, since it lets you explain what

is going on. It is also a good idea to add text to remind yourself at a later date what you

were doing and why.

To add text above a command, position the cursor anywhere in the command. Then

open the Insert menu, and then the Paragraph submenu, and choose Before Cursor. When

you type you will get ordinary text that will not be interpreted as part of the command.A.6 Printing
BEFORE YOU PRINT: You should clean up your session before you print, so that you do

not waste paper and so that the printed copy is attractive and easy to read. If you want

to print only part of the session, first save the session (see the section Saving your work)

and then cut out the portions of the session that you do not want to print. You might

want to save the trimmed version using a different name.

To print your worksheet, choose Print from the File menu, or click on the print button

(the fourth little button from the left in the menu bar).

If you would like to print just a plot, you need to display it in a window of its own.

You can do this by changing the plot display option, as described in the Plotting section.

Then position the cursor in the plot command and press return. The plot should open in

a new window, and you can use the Print menu in that window to print the plot.A.7 Saving your work
You can save your Maple worksheet so that you can come back to it at a later time. Choose

the Save or Save As. . . item from the File menu in Maple; a new window will pop up.

(Use Save As if you want to save your worksheet under a new name.) You may also save

a worksheet using the third button from the left in the menu bar.

When you reload a worksheet (with the Open item on the File menu or with the

second button from the left) Maple will not automatically recreate all the values created

in the worksheet. You may selectively re-execute commands in the worksheet, or you may

use the Execute item on the Edit menu to execute the entire worksheet.

There are (somewhat out of date) manuals for Maple in the lab. You can also get help ‘on

line’ from the Maple program. The help system is very good and always available as you

work in Maple.

?int If you know the command you need help on,

just type its name after a question mark.

You can browse through the entire help system or search it using the Help menu.




